



Arduino Protothreads [Tutorial]

Arduino protothreads, when to use them, how to use them, and why?

In this tutorial I'll show you, step by step, how to use protothreads in your Arduino programs.

First I'll give you a template that you can use for any protothread you create. Then you'll see more complex examples and how to use multiple protothreads at the same time.

Table of Contents



1. What are protothreads?
2. Install the Protothreads library for Arduino
3. Create your first Arduino Protothread
 - 3.1. A simple example
 - 3.2. Breaking down the code
 - 3.3. Arduino Protothread template
4. Important things to know about Protothreads
5. Multiple Protothreads in the same Arduino Program
 - 5.1. The code
6. A complete application with Arduino protothreads
 - 6.1. The Arduino code
7. Arduino protothreads or not: it's up to you

What are protothreads?

Explaining what are protothreads is quite hard without an example. If you don't grasp the idea behind protothreads right now, don't worry. Continue reading and get through the code examples, this will help you make more sense of it.

Protothreads allow you to create multiple "threads". Basically this is a C library,

which has also been packaged as a library for Arduino. The main goal of the library is to provide a simpler way to write programs for event-driven systems in memory constrained environments (like micro-controllers).

Note that you can't achieve true multi-threading with Arduino, all the code is executed in one thread, line by line. With protothreads you'll just have the "feeling" of multi-threading.

Why to use Arduino protothreads? Well, if you happen to have a giant state machine in your code, where you also want to execute multiple tasks at the same time, protothreads might be a nice alternative for you.

Install the Protothreads library for Arduino

Protothreads is a C library. If you download it [from the main protothreads website](#) (written by Adam Dunkels), it won't work directly because it's not packaged as an Arduino library.

Do the following steps to install the library:

- [Download the Arduino Protothreads library](#).
- Move the archive file into your Arduino > libraries folder.
- Unpack the archive file (using WinRAR, 7-Zip, ...)
- Restart your Arduino IDE, and that's it!

The Arduino Protothreads library is now installed. To test that, simply create a new program and include this line:

```
1. #include <pt.h>
2.
3. void setup() {}
4.
5. void loop() {}
```

Compile using Arduino Uno or Mega as the chosen board. If you get no error, the library is successfully installed! If, however, you have some errors at this point, go back to the installation bullet points and make sure you do each one correctly, and in the right order.

Create your first Arduino Protothread

A simple example

Let's start by writing a simple protothread. For this example we'll just redo the blink LED example, so it will be easy for you to understand.

Here's the code with protothreads:

```
1. #include <pt.h>
2. #define LED_1_PIN 9
3.
4. static struct pt pt1;
5.
6. static int protothreadBlinkLED1(struct pt *pt)
7. {
8.     static unsigned long lastTimeBlink = 0;
9.     PT_BEGIN(pt);
10.    while(1) {
11.        lastTimeBlink = millis();
12.        PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 1000);
13.        digitalWrite(LED_1_PIN, HIGH);
14.        lastTimeBlink = millis();
15.        PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 1000);
16.        digitalWrite(LED_1_PIN, LOW);
17.    }
18.    PT_END(pt);
19. }
20.
21. void setup() {
22.     pinMode(LED_1_PIN, OUTPUT);
23.     PT_INIT(&pt1);
24. }
25.
26. void loop() {
27.     protothreadBlinkLED1(&pt1);
28. }
```

Breaking down the code

```
1. #include <pt.h>
2. #define LED_1_PIN 9
```

You need to include <pt.h> to use the protothread library. Also we define the LED to be on the digital pin number 9.

```
1. static struct pt pt1;
```

Here we declare a protothread using the struct "pt".

```
1. static int protothreadBlinkLED1(struct pt *pt)
2. {
3.     ...
4. }
```

This is the function related to the blink LED functionality. When creating a function which uses a protothread, you need to give a pointer to a protothread as a parameter. You can also give more parameters to the function, you're not limited to one.

```
1. static unsigned long lastTimeBlink = 0;
```

We need to save the last time the LED blinked. But with protothreads, you can't use local variables. Those will not be preserved when the protothread blocks. So what we do here is using the "static" keyword, to tell the program to only create the variable once, and reuse it later. Note that with "static", the scope of the variable is not global though, it's restricted to the protothreadBlinkLED1() function.

```
1. PT_BEGIN(pt);
2. while(1) {
3.     ...
4. }
5. PT_END(pt);
```

We then have PT_BEGIN() to start the "thread", and PT_END() to finish it. You need to use those 2 lines every time you create a new "thread" with protothreads. As you notice here, there is an infinite loop in the function. It means we want this "thread" to always run. This while(1) is not mandatory, you could create a protothread that does one action and then exits immediately.

But if the thread always runs, then will it take all the resources of the Arduino micro-controller? Well, that's why there is PT_WAIT_UNTIL().

```
1. lastTimeBlink = millis();
2. PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 1000);
3. digitalWrite(LED_1_PIN, HIGH);
4. lastTimeBlink = millis();
5. PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 1000);
6. digitalWrite(LED_1_PIN, LOW);
```

PT_WAIT_UNTIL() takes 2 arguments: the protothread you're using, and a condition. The program will pause the execution of this thread until the condition is true. And that's where the magic of protothreads is: your global program won't be stopped here, only this thread will block.

Arduino Protothread template

Here's a template you can use to create your protothreads with Arduino.

```

1.  #include <pt.h>
2.
3.  // Create your protothread(s)
4.  static struct pt pt1;
5.
6.  // A protothread function
7.  static int protothreadFunction(struct pt *pt)
8.  {
9.      // Start a protothread
10.     PT_BEGIN(pt);
11.
12.     /* Your code inside this thread */
13.
14.     // Stop a protothread
15.     PT_END(pt);
16. }
17.
18. void setup() {
19.     // Init your protothread(s)
20.     PT_INIT(&pt1);
21. }
22.
23. void loop() {
24.     // Execute your protothread(s)
25.     protothreadFunction(&pt1);
26. }

```

For each new protothread – we’ll see later in this tutorial how to do that, you just need to duplicate the code you wrote for your first protothread.

Important things to know about Protothreads

Protothreads are great, and with Arduino they bring a new set of nice features to achieve something close from multi-threading, with a very low overhead.

However, you have to be aware of some important information and limitations:

- Basically a protothread overhead is only 2 bytes, so you don’t need to worry about hidden memory costs during the program execution.
- **You can’t use local variables inside protothreads.** Those are not preserved when the program blocks. Use global variables or “static” keyword instead.
- **The execution of your code inside a protothread must be fast** (between two PT_WAIT_UNTIL), or else you will block the rest of the program. Remember: true multi-threading doesn’t exist on Arduino, only one line of code is executed at a time. So, don’t use `delay()` or any other blocking function, ever.
- Call all your protothreads in your `loop()` function, as fast as possible (see point above).
- Under the hood, protothreads are just C macros. When you write a

protothread the code will be transformed into a switch statement, nothing more. Protothreads are just here to make your programming life simpler, but it won't do everything for you.


- To get more advanced knowledge about protothreads, [read the notes](#) from Adam Dunkels, the library author.

Multiple Protothreads in the same Arduino Program

If you chose to start using Arduino protothreads, it's certainly not to execute only one action. You want to achieve something close from multi-threading, with different actions happening at the same time.

Let's add 2 more protothreads to the previous example. For that we'll add some hardware components:

- LED 1 will blink every 1 second
- LED 2 will blink every 0.5 seconds
- and LED 3 will be ON when a push button is pressed



The code

```
1. #include <pt.h>
2. #define LED_1_PIN 9
3. #define LED_2_PIN 10
4. #define LED_3_PIN 11
5. #define BUTTON_PIN 5
6.
7. // Declare 3 protothreads
8. static struct pt pt1, pt2, pt3;
9.
10. // First protothread function to blink LED 1 every 1 second
11. static int protothreadBlinkLED1(struct pt *pt)
12. {
13.     static unsigned long lastTimeBlink = 0;
14.     PT_BEGIN(pt);
15.     while(1) {
16.         lastTimeBlink = millis();
17.         PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 1000);
18.         digitalWrite(LED_1_PIN, HIGH);
19.         lastTimeBlink = millis();
20.         PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 1000);
21.         digitalWrite(LED_1_PIN, LOW);
22.     }
23.     PT_END(pt);
24. }
25.
26. // Second protothread function to blink LED 2 every 0.5 second
27. static int protothreadBlinkLED2(struct pt *pt)
28. {
29.     static unsigned long lastTimeBlink = 0;
30.     PT_BEGIN(pt);
31.     while(1) {
32.         lastTimeBlink = millis();
33.         PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 500);
34.         digitalWrite(LED_2_PIN, HIGH);
35.         lastTimeBlink = millis();
36.         PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 500);
37.         digitalWrite(LED_2_PIN, LOW);
38.     }
39.     PT_END(pt);
40. }
41.
42. // Third protothread function to power on LED 3 if
43. // the push button is pressed.
44. static int protothreadPushButton(struct pt *pt)
45. {
46.     static unsigned long lastTimeCheck = 0;
47.     PT_BEGIN(pt);
48.     while (1) {
49.         lastTimeCheck = millis();
50.         PT_WAIT_UNTIL(pt, digitalRead(BUTTON_PIN) == HIGH);
51.         digitalWrite(LED_3_PIN, HIGH);
52.         PT_WAIT_UNTIL(pt, digitalRead(BUTTON_PIN) == LOW);
53.         digitalWrite(LED_3_PIN, LOW);
54.     }
55.     PT_END(pt);
56. }
57.
58. // In setup, set all LEDs as OUTPUT, push button as INPUT, and
59. // init all protothreads
60. void setup() {
```

```

61.   pinMode(LED_1_PIN, OUTPUT);
62.   pinMode(LED_2_PIN, OUTPUT);
63.   pinMode(LED_3_PIN, OUTPUT);
64.   pinMode(BUTTON_PIN, INPUT);
65.   PT_INIT(&pt1);
66.   PT_INIT(&pt2);
67.   PT_INIT(&pt3);
68. }
69.
70. // In the loop we just need to call the protothreads one by one
71. void loop() {
72.   protothreadBlinkLED1(&pt1);
73.   protothreadBlinkLED2(&pt2);
74.   protothreadPushButton(&pt3);
75. }

```

Once you know how to create and use one Arduino protothread, adding a few more won't be complicated. As you see here, for each protothread we just add a new function, we initialize the protothread in the `setup()`, and we run it in the `loop()`.


As you can see in the protothread for the push button, the condition inside the `PT_WAIT_UNTIL` is not about the time. Instead, we check if the button is pressed to continue the execution of the thread. This way, you get a very fast polling system for your button.


Note: as an exercise you could improve this code by [using interrupts](#). You would need to use a special pin for interrupt (ex: 2 or 3 for Arduino Uno), and set a flag as soon as the push button triggers an interrupt. In the protothread function you can check this flag and power on the LED accordingly.

A complete application with Arduino protothreads

In a [previous tutorial on Arduino multitasking](#) I explained how to achieve something close to multi-threading, while using nothing but the basic features of the C language. Here I'll rewrite the complete code used in this tutorial, but using protothreads.

As a reminder here's the schematics of the circuit:





What we want to do:

- Blink LED 1 every second.
- Read user input from Serial (number between 0 and 255) and write the data to LED 2.
- Power on LED 3 if the push button is pressed.
- Power on LED 4 if the potentiometer value is greater than 512.
- Print the potentiometer value via Serial every 2 seconds.

The Arduino code

[Click here](#) to see the complete code without protothreads. And now, here's the code using Arduino protothreads.

```

1.  #include <pt.h>
2.  #define LED_1_PIN 9
3.  #define LED_2_PIN 10
4.  #define LED_3_PIN 11
5.  #define LED_4_PIN 12
6.  #define POTENTIOMETER_PIN A0
7.  #define BUTTON_PIN 5
8.
9.  // Declare 5 protothreads
10. static struct pt pt1, pt2, pt3, pt4, pt5;
11.
12. // 1st protothread function to blink LED 1 every second
13. static int protothreadTask1(struct pt *pt)
14. {
15.     static unsigned long lastTimeBlink = 0;
16.     PT_BEGIN(pt);

```

```

17.     while(1) {
18.         lastTimeBlink = millis();
19.         PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 1000);
20.         digitalWrite(LED_1_PIN, HIGH);
21.         lastTimeBlink = millis();
22.         PT_WAIT_UNTIL(pt, millis() - lastTimeBlink > 1000);
23.         digitalWrite(LED_1_PIN, LOW);
24.     }
25.     PT_END(pt);
26. }
27.
28. // 2nd protothread function to write number from
29. // Serial to LED 2
30. static int protothreadTask2(struct pt *pt)
31. {
32.     static unsigned long lastTimeCheck = 0;
33.     PT_BEGIN(pt);
34.     while(1) {
35.         lastTimeCheck = millis();
36.         PT_WAIT_UNTIL(pt, millis() - lastTimeCheck > 5);
37.         if (Serial.available()) {
38.             int userInput = Serial.parseInt();
39.             Serial.println(userInput);
40.             if (userInput >= 0 && userInput < 256) {
41.                 analogWrite(LED_2_PIN, userInput);
42.             }
43.         }
44.     }
45.     PT_END(pt);
46. }
47.
48. // 3rd protothread function to power on LED 3 if
49. // the push button is pressed.
50. static int protothreadTask3(struct pt *pt)
51. {
52.     static unsigned long lastTimeCheck = 0;
53.     PT_BEGIN(pt);
54.     while(1) {
55.         lastTimeCheck = millis();
56.         PT_WAIT_UNTIL(pt, digitalRead(BUTTON_PIN) == HIGH);
57.         digitalWrite(LED_3_PIN, HIGH);
58.         PT_WAIT_UNTIL(pt, digitalRead(BUTTON_PIN) == LOW);
59.         digitalWrite(LED_3_PIN, LOW);
60.     }
61.     PT_END(pt);
62. }
63.
64. // 4th protothread function to power on LED 4 if
65. // the potentiometer value is greater than 512
66. static int protothreadTask4(struct pt *pt)
67. {
68.     static unsigned long lastTimeCheck = 0;
69.     PT_BEGIN(pt);
70.     while(1) {
71.         lastTimeCheck = millis();
72.         PT_WAIT_UNTIL(pt, analogRead(POTENTIOMETER_PIN) > 512);
73.         digitalWrite(LED_4_PIN, HIGH);
74.         PT_WAIT_UNTIL(pt, analogRead(POTENTIOMETER_PIN) <= 512);
75.         digitalWrite(LED_4_PIN, LOW);
76.     }
77.     PT_END(pt);
78. }
79.

```

```

80. // 5th protothread function to print the potentiometer
81. // value to Serial every 2 seconds
82. static int protothreadTask5(struct pt *pt)
83. {
84.     static unsigned long lastTimePrint = 0;
85.     PT_BEGIN(pt);
86.     while(1) {
87.         lastTimePrint = millis();
88.         PT_WAIT_UNTIL(pt, millis() - lastTimePrint > 2000);
89.         int potentiometerValue = analogRead(POTENTIOMETER_PIN);
90.         Serial.print("Potentiometer value: ");
91.         Serial.println(potentiometerValue);
92.     }
93.     PT_END(pt);
94. }
95.
96. // In setup, set all LEDs as OUTPUT, push button as INPUT, and
97. // init all protothreads
98. void setup() {
99.     Serial.begin(9600);
100.
101.     pinMode(LED_1_PIN, OUTPUT);
102.     pinMode(LED_2_PIN, OUTPUT);
103.     pinMode(LED_3_PIN, OUTPUT);
104.     pinMode(LED_4_PIN, OUTPUT);
105.     pinMode(BUTTON_PIN, INPUT);
106.     PT_INIT(&pt1);
107.     PT_INIT(&pt2);
108.     PT_INIT(&pt3);
109.     PT_INIT(&pt4);
110.     PT_INIT(&pt5);
111. }
112.
113. // In the loop we just need to call the protothreads one by one
114. void loop() {
115.     protothreadTask1(&pt1);
116.     protothreadTask2(&pt2);
117.     protothreadTask3(&pt3);
118.     protothreadTask4(&pt4);
119.     protothreadTask5(&pt5);
120. }

```

This code does the exact same thing as the code we wrote in the Arduino Multitasking tutorial.

Arduino protothreads or not: it's up to you

In this tutorial you've seen how to create and use multiple protothreads in your Arduino program.

Protothreads do not provide you with new functionalities. Instead, they allow you to write your code in a different way, which may be optimized when you have a lot of tasks to execute, or a state machine you want to simplify.

Now, using Arduino protothreads in your programs is really up to you. You might

use them everywhere, or not at all.

The most important thing here is not whether you want to use protothreads or not. The most important thing is that you are familiar with how to achieve multi-tasking with Arduino. You must first understand what is multi-threading and how to fake it with Arduino. This is the hard part. After that, using protothreads is almost just a matter of taste.